

悬链线理论

王勇 wangyong.seu@qq.com

2016 年 6 月 13 日

1 基本假设

索由高强钢丝集束而成，相对抗弯刚度很小，其受力特点可以认为是完全柔性。在自重和张力作用下分析其线形和力学参数时，基本假设如下：

- 索是理想柔性，既不能受压也不能受弯；
- 索的材料符合胡克定律；
- 索的横截面积在外荷载作用下的变化量十分微小，可忽略不计。

为了确定重力作用下索的线形，以弦左端点为原点、竖直向上为 Y 轴正方向建立右手直角坐标系。则索受到的重力沿 Y 轴负向。假设单位长度索的质量恒定，且不随张力变化。

2 符号约定

s 从索左端点（坐标系原点）开始计算的索长度；

μ 单位长度索的质量（假设是恒定的）；

$T(s)$ 索长度为 s 处的索张力（根据柔性索假设，张力沿索的切线方向）；

$T_y(s)$ 索长度为 s 处的索张力的 Y 向分量；

T_x 张力的 X 向分量（任取索段进行受力分析，由 X 向平衡方程可知 T_x 沿索长是均匀的）；

$\theta(s)$ 索切向量与 X 轴正向的夹角。




图 1: 索段计算示意图

3 索曲线求解

索段竖向的平衡方程为:

$$\begin{aligned}
 T_y(s) &= g \int_{s^0}^s \mu ds + T_y^0 \\
 T_x \tan(\theta(s)) &= g\mu \int_{s^0}^s ds + T_y^0 \\
 T_x \frac{dy}{dx} &= g\mu \int_{s^0}^s \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dx + T_y^0 \\
 \frac{d^2y}{dx^2} &= \frac{g\mu}{T_x} \sqrt{1 + \left(\frac{dy}{dx}\right)^2}
 \end{aligned} \tag{1}$$

此二阶微分方程的解析解为:

$$y = \frac{T_x}{g\mu} \cosh\left(\frac{g\mu}{T_x} + c_1\right) + c_2 \tag{2}$$

式中, c_1 和 c_2 为由边界条件确定的积分常量。代入边界条件: $x = 0, y = 0; x = L, y = C$ 得:

$$\begin{cases} \beta = \frac{g\mu L}{2T_x} \\ c_1 = \sinh^{-1}\left(\frac{\beta C/L}{\sinh \beta}\right) \\ c_2 = -\frac{T_x}{g\mu} \cosh(c_1) \end{cases} \tag{3}$$

悬链线索的形状长度 S 和无应力长度 S_0 分别为:

$$S = \frac{T_x}{g\mu} \left[\sinh \left(\frac{g\mu L}{T_x} + c_1 \right) + \sinh(c_1) \right] \quad (4)$$

$$\begin{aligned} S_0 &= S - \Delta S \\ &= S - \frac{T_x}{EA g\mu} \left[\frac{1}{2} g\mu L + \frac{1}{8} T_x (e^{2(c_1+2\beta)} - e^{-2(c_1-2\beta)} - e^{2c_1} + e^{-2c_1}) \right] \end{aligned} \quad (5)$$

4 算例

问题描述: 一根跨度为 $L = 888\text{m}$, 右侧支座比左侧支座低 80m , 跨中矢高 $f = 100\text{m}$, 每延米重 $g\mu = 60\text{kN/m}$ 。索的弹性模量 $E = 2 \times 10^5\text{MPa}$, 面积 $A = 0.6\text{m}^2$ 。假设在外荷载作用下两支座的间距及高差保持不变。试求如下参数: 索的水平力 T_x , 索的无应力长度 S_0 及索在受荷载作用下的形状长度 S 。

解题思路: 单缆在自重和张力作用下, 其线形为悬链线, 故可采用悬链线经典公式来进行求解。采用直接建模的方式, 单缆采用 LINK1 单元模拟, 单元水平长度为 1m 。分析时, 首先假定水平力大小, 根据悬链线方程求解节点坐标, 由此建立节点和单元, 并分析单缆在自重作用下的内力和线形。如果求解获得的水平力与假定水平力之间的误差较大或者单缆变形较大, 则返回重新计算, 直至满足误差要求。

! UNITS: m-kN

FINISH \$ /CLEAR \$ /PREP7

EA=0.6	! 单缆面积 (m^2)
EE=2.0E8	! 弹性模量 (kN/m^2)
QQ=60	! 单缆每延米重量 (kN/m)
YSM0=100	! 中跨矢高 (m)
CH=80	! 中跨锚点高差 (m)
SPAN=888	! 跨径

ET,1, LINK1	! 定义单元
R,1, EA,1.0E-5	! 设置一个较小的初应变
MP,EX,1, EE	! 定义材料特性
MP,PRXY,1, 0.3	
MP,DENS,1, 100	

*DIM,X,ARRAY,1000,1

*DIM,Y,ARRAY,1000,1

! 根据抛物线理论计算水平力迭代初始值

FF=YSM0-CH/2

HH=QQ*SPAN**2/(8*FF)	! 计算水平力
----------------------	---------

HH1=0.9*HH	! 设置水平力迭代区间
------------	-------------

HH2=1.1*HH

! 用二分法迭代计算主缆水平力

```

*DO, I, 1, 100, 1
    HFM=(HH1+HH2)/2                                ! 水平力迭代初值
    CI=QQ/HFM                                        ! 中间参数
    A0=CH*CI/SINH(SPAN*CI/2)/2                      ! 中间参数
    AI=LOG(A0+SQRT(A0**2+1))+SPAN*CI/2              ! 中间参数 alpha
    BI=COSH(AI)/CI                                   ! 中间参数 alpha1
    YSM=BI-COSH(CI*(SPAN/2)-AI)/CI                  ! 计算跨中矢高
    *IF , YSM, GT, YSM0, THEN                        ! 修正水平力
        HH1=HFM
    *ELSEIF , YSM, LE, YSM0
        HH2=HFM
    *ENDIF
*ENDDO

ERROR1=(YSM-YSM0)/YSM0                             ! 跨中处索垂度误差
MSS=(SINH(CI*SPAN-AI)+SINH(AI))/CI                 ! 按悬链线方程得出的形状长度
DSO=HFM*(SPAN+(SINH(2*CI*SPAN-2*AI)+SINH(2*AI))/2/CI)/(2*EE*EA)
                                                    ! 按悬链线方程得出的伸长量
YBM=DSO/(MSS-DSO)                                  ! 中跨的初始应变
R, 1, EA, YBM                                       ! 修改单缆初始应力
*DO, I, 1, 899, 1                                  ! 定义单缆节点
    X(I)=I-1
    Y(I)=BI-COSH(CI*X(I)-AI)/CI
    N, I, X(I), -Y(I), 0
*ENDDO

TYPE, 1 $ MAT, 1 $ REAL, 1 $ ESYS, 0              ! 单缆单元
*DO, I, 1, 888, 1
    E, I, I+1
*ENDDO
EPLLOT
FINISH

/SOLU
D, 1, ALL                                           ! 定义边界条件
D, 889, ALL
ACEL, 0, 1, 0                                       ! 施加重力荷载
ANTYPE, STATIC
NLGEOM, ON
NROPT, AUTO
LUMPM, OFF
EQLV, , , 0

```

```
AUTOTS,OFF
NSUBST,100
```

```
KBC,0
```

```
ALLSEL
SOLVE
```

```
/POST1
PLNSOL,U,Y
```




图 2: 单缆变形图

结果分析，如图 4所示，单缆在自重作用下的最大竖向位移为3.142 mm，相比于跨径，改变形足够小，认为已满足精度要求。此时求得的单缆力学参数如下：单缆水平力 $T_x = 99\,549.776\text{ kN}$ ，受荷后曲线长度为 $S = 902.1906\text{ m}$ ，单缆伸长量 $\Delta S = 0.7606\text{ m}$ ，无应力长度为 $S_0 = 901.4299\text{ m}$ 。